
MuPAD Light Version 2.5.3 — Examples

MuPAD is a computer algebra system originally developed in the early 1990’s at the Uni-
versity of Paderborn. It is now developed in cooperation with SciFace Software.

1. (Comments) To type a comment, start a line with two slashes.

• // It’s a great day to do math!

2. (Numeric Computation) Find the exact and approximate values of cos(π/5).

• num:=cos(PI/5)
51/2

4
+ 1/4

• float(num)

0.8090169944

• DIGITS:=25

25

• float(num)

0.8090169943749474241022934

• delete(DIGITS)

3. (Functions) Define f as a function and evaluate at the given point.

f(x) = 2 sin2 x cos x, x = π

• f:=x->2*sin(x)^2*cos(x)

x → 2 ∗ sin(x)ˆ2 ∗ cos(x)

• f(PI)

0

4. (Expressions) Define f as an expression and evaluate at the given point.

f(x) = 2 sin2 x cos x, x = π

• f:=2*sin(x)^2*cos(x)

2 cos(x) sin(x)2

• subs(f,x=PI)

2 cos(PI) sin(PI)2

• simplify(%)

0

5. (Graphing) Sketch the graphs of y = sin x and y = cos x for 0 ≤ x ≤ 2π.

• plotfunc2d(sin(x),cos(x),x=0..2*PI)

output omitted

6. (Inequalities) Find the solution set for the inequality x2 − 3x ≥ 4.

• solve(x^2-3*x>=4,x)

]− infinity, −1] union [4, infinity[

7. (Linear Systems) Solve the following system of equations.

2x + 3y − z = 9
−4x + y + 3z = 0

5x − 7y − 2z = −5

• solve({2*x+3*y-z=9,-4*x+y+3*z=0,5*x-7*y-2*z=-5},{x,y,z})

{[x = 121/36, y = 73/36, z = 137/36]}

8. (Limits) Compute the left- and right-hand limits at x = 2 for the function

f(x) =


x2, x < 2
3, x = 2
4− x, x > 2

.

• f:=piecewise([x<2,x^2],[x=2,3],[x>2,4-x])

piecewise(x2 if x < 2, 3 if x = 2, −x + 4 if 2 < x)

• limit(f,x=2,Left)

4

• limit(f,x=2,Right)

2

• limit(f,x=2)

undefined

9. (Continuity) Find all points at which the following rational function is discontinuous.

R(x) =
x− 2

x2 − 5x− 6

• discont((x-2)/(x^2-5*x-6),x)

{2, 3}

10. (Limit at Infinity) Evaluate: lim
x→∞

sin x

x
.

• limit(sin(x)/x,x=infinity)

0

11. (Derivative by Definition) Use the limit definition of the derivative to find g′(x) if
g(x) =

√
2x + 1.

• g:=x->sqrt(2*x+1)

x → sqrt(2x + 1)

• limit((g(x+h)-g(x))/h,h=0)
1

(2x + 1)1/2

12. (Derivative of an Expression) Find the slope of the line tangent to the graph of y = x+
2

x
at the point (1, 3).

• y:=x+2/x

x +
2

x

• dydx:=diff(y,x)

1− 2

x2

• subs(dydx,x=1)

−1

13. (2nd Derivative of a Function) Find the 2nd derivative of f(x) = x sin x.

• f:=x->x*sin(x)

x → x ∗ sin(x)

• f’’(x)

2 cos(x)− x sin(x)

14. (Graphing) Sketch the graph of y = sin(x)/x and label the x-axis in units of π.

• plotfunc2d(Ticks=[Steps=[PI,3],Automatic],sin(x)/x,x=0..20)

output omitted

15. (Linearization) Find the linearization, L(x), of the function f(x) =
√

x2 + 9 at x = −4.
Sketch the graph of both f and L near x = −4.

• f:=sqrt(x^2+9); df:=diff(f,x)

(x2 + 9)1/2

x

(x2 + 9)1/2

• x0:=-4; fx0:=subs(f,x=x0); dfx0:=subs(df,x=x0)

−4

5

−4/5

• L:=fx0+dfx0*(x-x0)

9/5− 4x

5

• plotfunc2d(f,L,x=-8..0)

output omitted

16. (Newton’s Method) Starting with x0 = 1, take five steps of Newton’s method to ap-
proximate a solution of x− cos x = 0.

• f:=x->x-cos(x)

x → x− cos(x)

• x:=1.0

1.0

• for i from 1 to 5 do x:=x-f(x)/f’(x); print(x) end_for:

0.7503638678

0.7391128909

0.7390851334

0.7390851332

0.7390851332

17. (Regression) Fit the following data to a model of the form y = A + B ln x + C/x2.

(1, 1), (4, 2), (11, 3), (31, 4), (83, 5), (227, 6)

• stats::reg([1,4,11,31,83,227],[1,2,3,4,5,6],A+B*ln(x)+C/x^2,[x],[A,B,C])

[[0.5971200686, 0.995516558, 0.4028178245], 0.0004391734841]

18. (Riemann Sums) Partition the interval [1, 5] into 20 subintervals of equal width and use
right endpoints to find the corresponding Riemann sum for the function f(x) = 1/x.

• export(student)

• f:=1/x
1

x

• riemann(f,x=1..5,20,Right)

sum

 1
i1

5
+ 1

, i1 = 1..20


5

• float(%)

1.532624844

• plotRiemann(f,x=1..5,20,Right)

plot::Group()

• plot(%)

output omitted

19. (Indefinite Integrals) Evaluate the indefinite integral

∫
2z dz

3
√

z2 + 1
.

• int(2*z/(z^2+1)^(1/3),z)
3z2

2
+ 3/2

(z2 + 1)1/3

20. (Partial Fraction Decomposition) Find the partial fraction decomposition of

R(x) =
1

x2 − x
.

• partfrac(1/(x^2-x))
1

x− 1
− 1

x

• normal(%)
1

x2 − x

21. (Finite Sums) Find the 100th harmonic number.

• sum(1/n,n=1..100)

output omitted

• float(%)

5.187377518

22. (Vector Operations) Let ~u = 3ı̂ + 2̂− k̂ and ~v = −3ı̂ + 7k̂. Find ~u · ~v, ~u× ~v, and the
angle between ~u and ~v.

• export(linalg)

• u:=matrix(3,1,[3,2,-1]); v:=matrix(3,1,[-3,0,7]) 3
2

−1


 −3

0
7


• scalarProduct(u,v)

−16

• crossProduct(u,v)  14
−18

6


• angle(u,v)

PI − arccos

(
4 141/2 581/2

203

)

23. (Partial Derivative) Find fx(1, π, 2) if f(x, y, z) = y sin(xyz).

• diff(y*sin(x*y*z),x)

y2z cos(xyz)

• subs(%,x=1,y=PI,z=2)

2 PI2 cos(2 PI)

• simplify(%)

2 PI2

• float(%)

19.7392088

24. (Higher-Order Partial Derivative) Compute fxxyz if f(x, y, z) = sin(3x + yz).

• f:=sin(3*x+y*z)

sin(3x + yz)

• diff(f,x,x,y,z)

9yz sin(3x + yz)− 9 cos(3x + yz)

25. (Implicit Differentiation) Given the equation xy + z3x − 2yz = 0, find ∂z/∂x at the
point (1, 1, 1).

• f:=x*y+z^3*x-2*y*z

xy − 2yz + xz3

• dzdx:=-diff(f,x)/diff(f,z)

− y + z3

3xz2 − 2y

• subs(dzdx,x=1,y=1,z=1)

−2

26. (Critical Points) Let V (y, z) = 108yz − 2y2z − 2yz2. Find all points for which Vy(y, z)
and Vz(y, z) are simultaneously zero.

• V:=108*y*z-2*y^2*z-2*y*z^2

108yz − 2yz2 − 2y2z

• solve({diff(V,y)=0,diff(V,z)=0},{y,z})

{[y = 0, z = 0], [y = 18, z = 18], [y = 0, z = 54], [y = 54, z = 0]}

27. (Gradient Vector) Compute ∇f at the given point.

f(x, y, z) = x2 + y2 − 2z2 + z ln x, (1, 1, 1)

• export(linalg)

• f:=x^2+y^2-2*z^2+z*ln(x)

z ln(x) + x2 + y2 − 2z2

• gradf:=grad(f,[x,y,z])  2x +
z

x
2y

−4z + ln(x)


• subs(gradf,x=1,y=1,z=1)  3

2
ln(1)− 4


• simplify(%)  3

2
−4

28. (Curvature) Find the curvature of ~r(t) = (2t− sin t)̂ı+(2−2 cos t)̂ at the point where
t = 3π/2.

• export(linalg)

• assume(t,Type::Real)

Type::Real

• r:=matrix[2,1,[2*t-sin(t),2-2*cos(t)])[
2t− sin(t)
−2 cos(t) + 2

]
• v:=diff(r,t) [

− cos(t) + 2
2 sin(t)

]
• mag_v:=norm(v,2)

(cos(t)2 − 4 cos(t) + 4 sin(t)2 + 4)1/2

• T:=v/mag_v

output omitted

• K:=norm(diff(T,t),2)/mag_v

output omitted

• curvature:=subs(K,t=3*PI/2)

output omitted

• simplify(curvature)

21/2

16

29. (Surface of Revolution) Sketch the graph of the surface obtained by rotating the graph
of y =

√
x, 0 ≤ x ≤ 1, about the x-axis.

• f:=sqrt(x)

x1/2

• plot(plot::xrotate(f,x=0..1))

output omitted

30. (Directional Derivatives) Find the directional derivative of

f(x, y) = 2xy − 3y2

at (5, 5) in the direction of ~u = 4ı̂ + 3̂.

• export(linalg)

• f:=2*x*y-3*y^2

2xy − 3y2

• u:=matrix(2,1,[4,3]) [
4
3

]
• gradf:=subs(grad(f,[x,y]),x=5,y=5)[

10
−20

]
• scalarProduct(gradf,u)/norm(u,2)

−4

31. (Ordinary Differential Equations) Solve the following differential equation:

xy′ + 2y = xex/2

• eq:=ode(x*y’(x)+2*y(x)=x*exp(x/2),y(x))

ode
(
2 y(x) + x diff(y(x), x)− x exp

(x

2

)
, y(x)

)
• solve(eq) (

c1

x2
+ 2 exp(x)1/2 − 8 exp(x)1/2

x
+

16 exp(x)1/2

x2

)

32. (Initial Value Problems) Solve the following initial value problem:

y′′ + 2y′ + y = e2x; y(0) = 2, y′(0) = 1

• eq:=ode({y’’(x)+2*y’(x)+y(x)=exp(2*x),y(0)=2,y’(0)=1},y(x))

ode({y(0) = 2, D(y)(0) = 1, y(x) + · · · − exp(2x)}, y(x))

• solve(eq) (
17 exp(−x)

9
+

exp(x)2

9
+

8x exp(−x)

3

)

9



33. (Laplace Transforms) Find the Laplace transform of f(t) = t2 − 1 + cos t.

• export(transform)

• laplace(t^2-1+cos(t),t,s)
2

s3
− 1

s
+

s

s2 + 1

• invlaplace(%,s,t)

cos(t) + t2 − 1

34. (One Variable Statistics) Find the mean, median, mode, and standard deviation of the
data set given below.

{92.5, 43, 78, 82, 57.5, 63, 78, 91, 84.5, 68}

• export(stats)

• data:=[92.5,43,78,82,57.5,63,78,91,84.5,68]

[92.5, 43, 78, 82, 57.5, 63, 78, 91, 84.5, 68]

• mean(data)

73.75

• median(data)

78

• modal(data)

[78], 2

• stdev(data)

15.73080841

• stdev(data,Population)

14.92355521

35. (Surface Plots) Sketch the surface defined by z = sin xy.

• plotfunc3d(sin(x*y),x=-PI..PI,y=-PI..PI,Grid=[200,200])

output omitted

36. (Parametric Plots) Sketch the space curve defined by the parametric equations.

x = sin t, y = cos t, z = t

• p1:=plot::Curve3d([sin(t),cos(t),t],t=0..4*PI)

plot::Curve3d([sin(t), cos(t), t], t = 0..4 PI)

• plot(p1)

output omitted

37. (Matrix Operations) Given the two matrices A and B as shown below, find A + 2B,
AT , AT A, and B−1.

A =

 1 0 4
3 2 −2
0 −1 2

 B =

 1 1 1
3 0 3

−1 −3 −5


• A:=matrix(3,3,[[1,0,4],[3,2,-2],[0,-1,2]]) 1 0 4

3 2 −2
0 −1 2


• B:=matrix(3,3,[[1,1,1],[3,0,3],[-1,-3,-5]]) 1 1 1

3 0 3
−1 −3 −5


• A+2*B  3 2 6

9 2 4
−2 −7 −8


• C:=linalg::transpose(A)  1 3 0

0 2 −1
4 −2 2


• C*A  10 6 −2

6 5 −6
−2 −6 24


• B^(-1)  3/4 1/6 1/4

1 −1/3 0
−3/4 1/6 −1/4

38. (Gauss-Jordan Elimination) Use Gauss-Jordan elimination to solve the following sys-
tem of equations.

2x + 3y − z = 9
−4x + y + 3z = 0

5x − 7y − 2z = −5

• M:=matrix(3,4,[[2,3,-1,9],[-4,1,3,0],[5,-7,-2,-5]]) 2 3 −1 9
−4 1 3 0

5 −7 −2 −5


• linalg::gaussJordan(M)  1 0 0 121/36

0 1 0 73/36
0 0 1 137/36


39. (Nonlinear Equations) Sketch the graphs of the following equations and numerically

approximate a point of intersection of the graphs.

4x2 + y2 − 4 = 0

x + y − sin(x− y) = 0

• f:=4*x^2+y^2-4; g:=x+y-sin(x-y)

4x2 + y2 − 4

x + y − sin(x− y)

• p1:=plot::implicit(f=0,x=-2..2,y=-2..2)

plot::Group()

• p2:=plot::implicit(g=0,x=-2..2,y=-2..2)

plot::Group()

• plot(p1,p2)

output omitted

• numeric::solve({f=0,g=0},{x=1,y=0})

{[y = −0.1055304923, x = 0.9986069441]}

40. (ODE Direction Fields) Sketch the direction field associated with the differential equa-
tion

dy

dx
= −y

x

• p1:=plot::vectorfield([1,-y/x],x=-10..10,y=-10..10)

plot::Group()

• p2:=plot::Scene(p1,Scaling=Constrained,Title="dy/dx=-y/x")

plot::Group()

• plot(p2)

output omitted

